
Course Notes


Algorithm = how to do something (not necessarily in programming language, could be in English), expressed as a detailed list of instructions for accomplishin a task.

Properties:

Well-defined input (may be nothing)

Unambiguous sequence of instructions

Instructions are atomic (indivisible)

Produces some outcome

Result is either the transformation of input OR physical changes

Terminates when task is complete


How good is an algorithm?

How fast is it?

Does it respond to every input?

Does it waste resources ie. memory?

Is it easy to communicate?


Is everything an algorithm? No. 

	Algorithm
	Software

	Specific function/goal
	Tool, not designed for specific function



Programming

Sometimes implementing algorithms
Devising new algorithms

Sometimes implementing behaviours, not algorithms

Event-driven programming software responds to outside events (mouse click)

Language Syntax

What instructions are available

How to format an instruction
Special Symbols

Semicolon
 ; used to end a statement

Parentheses()-indicates input to a function

Curly braces{}- used to group statements

Variables

Names given to data with a specific meaning

Some are already given in Processing: height (of screen), width, mouseX, mousey
Value can change as the program executes but the meaning stays the same (ex: score in a hockey game, bank balance)

Iteration (looping)

Used to do something over and over
Processing has a “draw” loop (“setup” function called only once at the beginning, “draw” function called after, loops at 30 fps)

Processing Structure

void setup()

{

size(400,400);
}
void draw()
}

ellipse(20,40,60,80);
}

Conditionals
Execute or do not execute some code based on current state

Eg. automatic lawn sprinklers do not turn on in the rain

Functions

A way to organize and re-use code

Refer to a process by name, eg. ellipse()
Example:

void centralCircle()

{


ellipse(width/2,height/2,40,40);

}

Text Setup

void setup()

{

Pfont f;

f = create Font(“Arial”,48);

textFont(f);

}

void draw()

{

text(“hello world”,10,100);

}

Expressions – something that can be evaluated to produce a value ex: 1+2*3
Variables – a named place to store a value

names should mean something

may include numbers

no spaces/punctuation

may not start with a number

certain reserved words

can only store a certain type of data (specify type when declaring variable)

Types


Primitive types

int – integer, whole #

float – floating point value (decimal)

char – single character, ie. “a”

boolean – true/false

long – very large integer


Composite types

invented by programmers

seem to be built in

ex. string – text data

PImage – processing image

PFont – processing font

Casting
int x = 3.2 ( WRONG TYPE

float x = 3 //OK

int y = 3*x  // WRONG!! “float” made x incompatible w/int

//instead…

int y =(int)(3*x) //This is casting, saying you are ok w/discarding info (rounding down)

In other words… (int)3.9 == 3 (throws away decimal)

Scope – a variable is known only inside the function where you declared it, ie. setup and draw. “There’s more than one Main Street!”. It’s good practice to declare variables before setup/draw.

Conditionals and Logic
If Loop

if (x > 50)

{


ellipse(20,20,20,20);

}

And

((x>50) && (x<150)) //Both must be true

Not
(!(x>50)) //If x is NOT greater than 50 (negates the following expression

Or

((x > 50) || (x < 150)) //At least one must be true

If… Else

if (x > 50)

{

ellipse (mouseX,mousey,20,20);

}

else

{

//something else…

}

While Loop

int i = 0;

while (i < 50) //Test i, done yet?

{

text(i,200,200);

i++; //Always increment

}

For Loop

for when you know in advance how many iterations you need

for(int i = 0; i < 50; i++)

{

something;

}

Remember to always reinitialize value (ie. x-value) at top of draw loop!

Arrays

Declaration: int[]myArray = new int [9] //where 9 is max size. (elements 0-8)

Intialization: use a for loop

for (int i = 0; i < myArray.length; i++)

{


myArray[i]=(int)random(width);


//or whatever the value should be

}

Array Processing


Suppose you have data in an array, you want summary info (eg. average). What do you do?

go through elements one by one, summing as you go
divide <array name>.length


Restricted Sum

eg. only passing grades

use if statement (ie. if x > 50 then sum)

remember to divide not by total but # of elements that fit criteria


EXAMPLES ---------------------------------------------------------------------------------------------------------
Beetle[] b;
color[] mycolors;
int pos[] = new int[3];

int numbeetles = 60;

void setup()
{
  size(600,400);
  background(255);
  b = new Beetle[numbeetles];
  
  // create all beetles in a pile at the centre of the screen
  for (int i = 0; i < numbeetles; i++)
  {
    b[i] = new Beetle(width/2,height/2);
  }
}

void draw()
{
  background(255);
  
  // loop over all beetles
  for (int i = 0; i < numbeetles; i++)
  {
    // draw this beetle
    fill(b[i].c);
    ellipse(b[i].x, b[i].y,b[i].sz,b[i].sz);
    if (!b[i].dead) // if beetle is alive
    {
       // beetle scurries: update position and movement variables
       b[i].x += b[i].speed*cos(b[i].direction*PI/180);
       b[i].y += b[i].speed*sin(b[i].direction*PI/180);
    
       b[i].direction = b[i].originaldirection + (int)random(-60,60);
       b[i].speed += (0.05)*random(-2,3);
    }
    float dx, dy;
    dx = mouseX-b[i].x;
    dy = mouseY-b[i].y;
    // are we close to this beetle and pressing the button?
    if (dx*dx+dy*dy < 1000 && mousePressed)
    {
      // squish the beetle
      b[i].dead = true;
      b[i].c = color(255,0,0);
      mousePressed = false; // reset mousePressed until next time
    }
  }
}
class Beetle
{  
  float x;
  float y; // current coordinates

  int ID; // an ID number (not used yet)
  float speed; // how fast it is running
  int direction; // angle of current heading
  int originaldirection; // beetle's preferred heading    
  color c; // beetle color
  int sz; // beetle size
  boolean dead; // dead or alive?
  
  // constructor:
  Beetle(float tx, float ty)
  {
    dead = false;
    x = tx;
    y = ty;
    ID = 0;
    speed = random(1,3);
    speed = speed*0.1;
    direction = (int)random(0,359);
    originaldirection = direction;
    c = color(random(255),random(255),random(255));
    sz = 20;
  }  
}
Linear Search Function

int number[];
PFont f;
int margin = 110;
void setup()
{
  number = new int[10];
  for (int i = 0; i < 10; i++)
    number[i] = (int)random(100);
  f = createFont("Aharoni-Bold",64);    
  size(1200,800);
  textFont(f);
}
void draw()
{
 String s;
 background(0); 
 int total = 0;
 int count = 0;
  for (int i = 0; i < 10; i++)
  {
    fill(100);
    rect(margin+i*90,90,77,77);
    text(i,margin+16+i*90,90);
    fill(255);
 
    s = "";
    if (number[i]<10) s += " "; // formatting
    
    text(s+number[i],margin+i*90,155);
  }   text(LinearSearch(number,number.length),500,500); 
}
float LinearSearch(int[] inarray, int sz)
{

  // finds the biggest entry among the first sz elements of the input array

  int biggest;
  
  biggest = inarray[0];
  
  for (int i = 0; i < sz; i++)
  {
    if (biggest < inarray[i])
       biggest = inarray[i];
  }  
  return biggest;
}

Credit Card Interest

// note, make sure to create the font before running

 PFont f;
 
 void setup()
 {
   size(400,400);
   f = loadFont("Serif-48.vlw");
   textFont(f);
   frameRate(1);
 } 
 void draw()
 {  
  background(255);
  fill(0);
   float p = 100;
   float interest = 0.18;
   
   for (int i = 0; i < 12; i++)
   {
//     text(p,20,i*36);
     p = p + (interest*p)/12.0;
   }
   
   fill(0);
   text(p,20,120);
 }

Curiosity Behaviour

int x = 200;
int y = 200;

int d = 0;

void setup()
{
  size(400,400);
  background(0);
}

void draw()
{
  ellipse(x,y,20,20);
  int dx = x - mouseX;
  int dy = y - mouseY;
  d = (int)sqrt(dx*dx+dy*dy);
  
  if (d > 100)
  {
    // chase
    if (x > mouseX)
  {
    x--;
  }
  else
  {
    x++;
  }
  
  if (y > mouseY)
  {
    y--;
  }
  else
  {
    y++;
  } 
  }
  else
  {
    // flee
  if (x < mouseX)
  {
    x--;
  }
  else
  {
    x++;
  }
  
  if (y < mouseY)
  {
    y--;
  }
  else
  {
    y++;
  }   
  }
  
 
}

Bounce, Gravity

float x, y;
float vx = 1.1;
float vy = 3;


void setup()
{
  size(400,400);
}

void draw()
{
  x = x + vx;  
  y = y + vy; // change position according to velocity
  vy = vy + 0.3; // increase vy by gravity
  
  if (y > height)
  {
    vy = vy*-0.8; // reverse direction, but less energy
    y = height; // doesn't get stuck in the floor
  }
  ellipse(x,y,20,20);
  
  
}

Note: “Void” means the function does not return information or compute a value.








